
Fundamental Discrete Structures

Goal: to reason about ordered pairs, options, sets, relations, functions,

and maps as fundamental data types in computer science, e.g., for

modeling and verification of digital systems.

• ordered pairs

• options

• sets, relations, and functions

• maps

Ordered pairs

• Intuitively, an ordered pair of two objects a and b, often written

in conventional notation as (a, b), is a pair of a and b in which

order is significant:

• a is the first element of the pair and

• b is the second.

• It follows that, if a 6= b, then (a, b) and (b, a) are two distinct

ordered pairs, even though they contain the same elements.

• Indeed, the fundamental property of ordered pairs is that (a1, b1)

is the same pair as (a2, b2) iff a1 = a2 and b1 = b2.

• When we formalize ordered pairs as an algebraic datatype, this

property follows directly from the datatype’s axioms.

Ordered pairs

• We introduce ordered pairs with a polymorphic datatype Pair,

which has only one constructor, the binary symbol pair.

• We also declare two selectors, one for retrieving the left and

another for retrieving the right element of an ordered pair:

datatype (Pair S T) := (pair pair-left:S pair-right:T)

• All of the code for ordered pairs is introduced inside a predefined

module Pair that is loaded when Athena first starts:

module Pair {

...

}

• Because ordered pairs are so widely used, this module is by

default opened at the top level.

Ordered pairs

• We introduce the symbol @ as an alias for the constructor pair,

so we can write pairs (a, b) in infix notation as (a @ b).

• We want the pair constructor to bind quite tightly, so we give it a

high precedence level:

set-precedence pair 310

define @ := pair

• Occasionally we also want to write ordered pairs using

square-bracket notation, writing [1 2] for (pair 1 2), for

example.

• Conversely, we will want to transform (pair s t) terms into

two-element Athena lists of the form [s t].

• The procedures lst->pair and pair->lst let us do both.

Ordered pairs

• Finally, we define some variable abbreviations and assert the

relevant datatype and selector axioms:

define [x y z a b w p p1 p2] :=

[?x ?y ?z ?a ?b ?w ?p:(Pair 'S 'T) ?p1:(Pair 'S1 'T1)

?p2:(Pair 'S2 'T2)]

assert pair-axioms :=

(datatype-axioms "Pair" joined-with selector-axioms "Pair")

• No-confusion, no-junk axioms, as well as axioms for pair-left

and pair-right.

Ordered pairs

Let us now prove a few theorems:

conclude pair-theorem-1 :=

(forall p . p = (pair-left p) @ (pair-right p))

datatype-cases pair-theorem-1 {

(p as (pair x y)) =>

(! chain [p = (( pair-left p) @ (pair-right p)) [pair-axioms ]])

}

conclude pair-theorem-2 :=

(forall x y z w . x @ y = z @ w <==> y @ x = w @ z)

pick-any x y z w

(! chain [(x @ y = z @ w) <==> (x = z & y = w) [pair-axioms]

<==> (y = w & x = z) [prop-taut]

<==> (y @ x = w @ z) [pair-axioms ]])

Ordered pairs

We introduce a swap function that swaps the elements of an ordered

pair:

declare swap: (S, T) [(Pair S T)] -> (Pair T S)

assert* swap-def := [(swap x @ y = y @ x)]

> (eval swap swap 1 @ 'a)

Term: (pair 1 'a)

> (eval swap swap swap 1 @ 'a)

Term: (pair 'a 1)

conclude swap-theorem-1 :=

(forall x y . swap swap x @ y = x @ y)

pick-any x y

(! chain [(swap swap x @ y) = (swap y @ x) [swap-def]

= (x @ y) [swap-def ]])

Options

Option values are generated by the following polymorphic datatype:

datatype (Option S) := NONE | (SOME option-val:S)

Intuitively, this definition says that a value of sort (Option S) is either

• NONE or

• (SOME v) for some value v of sort S.

In the latter case, applying the selector function option-val to

(SOME v) retrieves the value v.

The symbol SOME is given low precedence (110), as we want its scope

to extend as far to the right as possible, as illustrated below.

> (SOME 'foo :: nil)

Term: (SOME (:: 'foo

nil:(List Ide )))

Options

Options are useful in a number of contexts, most notably for error

handling:

• If a function f that is normally supposed to produce a value of

sort S generates an error, it can instead return NONE as its output;

that will explicitly indicate the occurrence of an error.

• If, on the other hand, all goes well and f successfully computes a

result v of sort S, it can return the output (SOME v), indicating

that the computation went well and resulted in v.

• Hence, to make f deal with errors explicitly, we modify its

output sort by lifting it from S to (Option S).

Options

It is important to keep in mind the datatype axioms for Option:

1. NONE is distinct from any value of the form (SOME v);

2. (SOME v1) is identical to (SOME v2) iff v1 is identical to v2; and

3. every option value is either equal to NONE or to a value of the

form (SOME v).

For reference convenience, we give the name opt-axioms to the list of

these axioms:

assert opt-axioms := (datatype-axioms "Option")

define [opt-no-confusion opt-some-injective opt-no-junk] := opt-axioms

Options

Lemmas derived inside Options module opened at the top level.

define [x y z] := [?x ?y ?z]

conclude option-lemma-1 := (forall x y . x = SOME y ==> x =/= NONE)

pick-any x y

assume hyp := (x = SOME y)

(!chain- > [true ==> (NONE =/= SOME y) [opt-no-confusion]

==> (NONE =/= x) [hyp]

==> (x =/= NONE) [sym ]])

conclude option-lemma-2 :=

(forall x . x =/= NONE ==> exists y . x = SOME y)

pick-any x

assume hyp := (x =/= NONE)

(!dsyl (! uspec opt-no-junk x) hyp)

define option-lemma-2-conv :=

(forall x . (forall y . x =/= SOME y) ==> x = NONE)

Options

define option-lemma-3 :=

(forall x . x = NONE ==> ~ exists y . x = SOME y)

define option-lemma-4 :=

(forall x y . x = NONE ==> x =/= SOME y)

conclude option-lemma-5 :=

(forall x y z . x = SOME y & y =/= z ==> x =/= SOME z)

pick-any x y z

assume h := (x = SOME y & y =/= z)

(!chain- > [h ==> (y =/= z) [right-and]

==> (SOME y =/= SOME z) [opt-some-injective]

==> (x =/= SOME z) [h]])

define opt-lemmas := [option-lemma-1 option-lemma-2 option-lemma-2-conv

option-lemma-3 option-lemma-4 option-lemma-5]

define option-results := (join opt-axioms opt-lemmas)

Sets

Finite sets are introduced through a polymorphic structure (Set S)

that is inductively generated from two constructors:

• a nullary constructor null, representing the empty set of sort S,

and

• a binary constructor insert, which adds an element of sort S into

a set of sort (Set S).

structure (Set S) := null | (insert S (Set S))

This structure definition, and all subsequent code and proofs are in an

Athena module named Set, part of the standard Athena library.

Unlike Options and Pair, this module is not open by default.

Sets

Observe that the Set definition appears to be structurally identical to

that of finite lists:

datatype (List S) := nil | (:: S (List S))

with the roles of nil and :: played by null and insert, respectively.

There is one crucial difference, however:

• Lists are freely generated (hence the datatype keyword),

whereas

• Sets are only inductively generated (hence the structure

keyword).

Free generation means that two lists are equal iff they are identical as

terms. Thus, (:: 1 (:: 2 nil)) and (:: 2 (:: 1 nil)) represent

different lists.

Sets

Free generation is reflected in the datatype-axioms of List, one of

which is:

(forall h1 t1 h2 t2 . h1 :: t1 = h2 :: t2 ==> h1 = h2 & t1 = t2)

and whose contrapositive is:

(forall h1 t1 h2 t2 . h1 =/= h2 | t1 =/= t2 ==> h1 :: t1 =/= h2 :: t2)

By contrast, the analogous axiom for sets does not hold.

Indeed, we will have to ensure that (insert 1 (insert 2 null)) and

(insert 2 (insert 1 null))

are identical in order to respect the extensionality principle, which

states that two sets are equal iff they contain the same elements.

Sets

It will be convenient to be able to transform square-bracket-enclosed

Athena lists into sets and vice versa, so we can write, for instance,

[1 2 3] as a shorthand for the set

(1 insert 2 insert 3 insert null).

The Set module introduces two such procedures, alist->set and

set->alist (the latter also removes duplicates from the final list.)

We expand the input space of insert so that it can accept Athena lists

as second arguments:

expand-input insert [id alist- >set]

And in the interest of brevity and readability, we define ++ as an alias

for insert:

define ++ := insert

Sets

We can now write sets in a succinct and fairly natural notation:

> (1 ++ [2 3])

Term: (insert 1

(insert 2

(insert 3

null:(Set Int ))))

We also set the precedence of ++ to be relatively high, and define some

abbreviations for term variables: x, y, z and h, h’ as polymorphic

variables of completely general sorts, and the rest specifically as

(polymorphic) set variables:

set-precedence ++ 210

define [x y z h h' s s' t t' s1 s2 s3 A B C D E] :=

[?x ... ?s:(Set 'T1) ... ?t:(Set 'T3) ... ?A:(Set 'T8) ...]

Set membership

The first relation we introduce is the most fundamental one: the set

membership relation.

We expand its input space so that it can take square-bracket-enclosed

lists as second arguments (which will be converted to set terms via

alist->set).

And we define the relation recursively, much in the same way that a

similar relation for list membership would be defined:

declare in: (S) [S (Set S)] -> Boolean [[id alist- >set]]

assert* in-def :=

[(~ _ in null)

(x in h ++ t <==> x = h | x in t)]

define [in-not-in-null in-def-insert] := in-def

Set membership

The following two lemmas are simple but handy:

conclude null-characterization := (forall x . x in [] <==> false)

pick-any x

(! equiv

assume hyp := (x in [])

(! absurd hyp (! uspec in-not-in-null x))

assume false

(! from-false (x in [])))

conclude in-lemma-1 := (forall x A . x in x ++ A)

pick-any x A

(!chain- > [(x = x) ==> (x in x ++ A) [in-def-insert ]])

We often use upper-case shorthands for characterization theorems:

define NC := null-characterization

We call this a “characterization” result for null, by the way, because it

provides necessary and sufficient membership conditions for null.

The subset relation

• We continue with another fundamental relation, built on top of

the membership relation: the subset relation, customarily

represented by the symbol ⊆.

• A set A is a subset of a set B iff every element of A is also an

element of B:

∀ A,B . A ⊆ B ⇔ (∀ x . x ∈ A ⇒ x ∈ B). (1)

• We could define subset directly by asserting (1). However, such

a definition would have a drawback:

• It would not be constructive, due to the unrestricted universal

quantifier that occurs on the right-hand side of the

biconditional.

The subset relation

• Without a constructive definition, typically given by recursion,

evaluation will not work; for example, eval will not be able to

reduce a term like ([1] subset [2 1 3]) to true.

• We want eval to be able to handle such terms, not only in order

to test our axioms, but also in order to test conjectures.

• For that reason, we will instead give a constructive definition of

subset. Once we have such a definition, we will then derive (1)

from it.

• This is a general strategy that we use, for instance, instead of

defining set-theoretic union by directly asserting that an element

x belongs to the union of A and B iff it belongs to A or it

belongs to B, we instead give a recursive definition of union and

then inductively derive the said property from it.

The subset relation

• In general, when there is a choice between a constructive

definition D1 and a nonconstructive definition D2, we will

typically prefer D1 and then proceed to derive D2 from D1,

usually by induction.

• In this case the required recursive definition can be given as:

declare subset: (S) [(Set S) (Set S)] -> Boolean [[alist- >set alist- >set]]

assert* subset-def :=

[([] subset _)

(h ++ t subset A <==> h in A & t subset A)]

> (eval [1 2] subset [3 2 4 1 5])

Term: true

> (eval [1 2] subset [3 2])

Term: false

The subset relation

• We now go on to derive the nonconstructive characterization of

the subset relation from this definition.

• This is done in two steps, beginning with the left-to-right

direction stating that if A is a subset of B (according to

subset-def) then for all x, if x belongs to A then x also belongs

to B. It is proved by structural induction, with basis case:

define subset-characterization-1 :=

by-induction (forall A B . A subset B ==> forall x . x in A ==> x in B)

{

null => pick-any B

assume (null subset B)

pick-any x

(! chain [(x in null) ==> false [NC]

==> (x in B) [prop-taut ]])

| (A as (insert h t)) => ...

}

The subset relation

and inductive step:

| (A as (insert h t)) =>

pick-any B

assume hyp := (A subset B)

pick-any x

let {IH := (forall B . t subset B ==>

forall x . x in t ==> x in B);

_ := (!chain- > [hyp ==> (t subset B) [subset-def ]])}

assume (x in A)

(! cases (!chain <- [(x = h | x in t)

<== (x in A) [in-def ]])

assume (x = h)

(!chain- > [hyp ==> (h in B) [subset-def]

==> (x in B) [(x = h)]])

(! chain [(x in t) ==> (x in B) [IH]]))

The subset relation

The converse direction is likewise proved by structural induction. With

both directions available, the classical (nonconstructive) subset

characterization is easy to derive.

define subset-characterization-2 :=

(forall A B . (forall x . x in A ==> x in B) ==> A subset B)

define subset-characterization :=

(forall s1 s2 . s1 subset s2 <==> forall x . x in s1 ==> x in s2)

We define SC as a handy abbreviation for subset-characterization:

define SC := subset-characterization

The subset relation

It will be very useful to have an introduction method for goals of the

form (s subset t), based on subset-characterization. The

method will take a premise of the form

(forall x . x in s ==> x in t)

and will use SC to derive the desired goal (s subset t):

define subset-intro :=

method (p)

match p {

(forall x ((x in A) ==> (x in B))) =>

(!chain- > [p ==> (A subset B) [SC]])

}

CSCI.6964/4964 Software Verification—Fundamental Proof Methods in Computer Science (Arkoudas and Musser)—Sections 10.1-10.3.3 – p. 27/62



Set identity

Recall that when an inductively generated structure is not free, we need

to define the identity relation on that structure; that is, we need to state

exactly when two terms of that structure are to count as identical. In

this case, the identity relation on sets is defined as should be expected:

Two sets are equal iff each is a subset of the other.

assert* set-identity := (A = B <==> A subset B & B subset A)

Recall also that when a structural identity axiom of this kind is

asserted, Athena automatically adjusts eval to ensure that it adheres to

the new definition. Thus, we now have:

> (eval 1 ++ 2 ++ [] = 2 ++ 1 ++ [])

Term: true

If we did not have a constructive definition of subset, we would also

not have a constructive set identity, and the eval call above would fail.

Set identity

An alternative and nonconstructive characterization of set identity is:

define set-identity-characterization :=

(forall A B . A = B <==> forall x . x in A <==> x in B)

define SIC := set-identity-characterization

We can readily derive SIC from set-identity and SC.

Similar to subset-intro, we define a method set-identity-intro

that takes two premises of the form (s subset t) and (t subset s)

and derives (s = t):

define set-identity-intro :=

method (p1 p2)

match [p1 p2] {

[(A subset B) (B subset A)] =>

(!chain- > [p1 ==> (p1 & p2) [augment]

==> (A = B) [set-identity ]])

}

Set identity

A more direct introduction method, based on SIC, is also useful.

This one takes a premise of the form

∀x . x ∈ A⇔x ∈ B

and derives the conclusion A = B:

define set-identity-intro-direct :=

method (premise)

match premise {

(forall x ((x in A) <==> (x in B))) =>

(!chain- > [premise ==> (A = B) [SIC ]])

}

Empty set characterizations

Using SIC we can also derive a couple of handy alternative

characterizations of the empty set:

conclude NC-2 := (forall A . A = null <==> forall x . ~ x in A)

pick-any A

(! chain [(A = null)

<==> (forall x . x in A <==> x in null) [SIC]

<==> (forall x . x in A <==> false) [NC]

<==> (forall x . ~ x in A) [prop-taut ]])

define NC-3 := (forall A . A =/= null <==> exists x . x in A)

The following will be a convenient abbreviation, defined as a

procedure:

define (non-empty s) := (s =/= null)

Subset relation properties

Some standard results about the subset relation can be readily derived

now: reflexivity, antisymmetry, and transitivity.

define subset-reflexivity := (forall A . A subset A)

define subset-antisymmetry :=

(forall A B . A subset B & B subset A ==> A = B)

conclude subset-transitivity :=

(forall A B C . A subset B & B subset C ==> A subset C)

pick-any A B C

assume (A subset B & B subset C)

(! subset-intro

pick-any x

(! chain [(x in A)

==> (x in B) [SC]

==> (x in C) [SC]]))

Proper subset relation

The proper subset relation is defined as follows:

declare proper-subset: (S) [(Set S) (Set S)] -> Boolean

[[alist- >set alist- >set]]

assert* proper-subset-def :=

[(s1 proper-subset s2 <==> s1 subset s2 & s1 =/= s2)]

> (eval [1 2] proper-subset [2 3 1])

Term: true

Proper subset relation

A characterization theorem and some useful lemmas follow:

define proper-subset-characterization :=

(forall s1 s2 . s1 proper-subset s2 <==>

s1 subset s2 & exists x . x in s2 & ~ x in s1)

define PSC := proper-subset-characterization

define proper-subset-lemma :=

(forall A B x . A subset B & x in B & ~ x in A ==> A proper-subset B)

define in-lemma-2 := (forall h t . h in t ==> h ++ t = t)

define in-lemma-3 := (forall x h t . x in t ==> x in h ++ t)

define in-lemma-4 :=

(forall A x y . x in A ==> y in A <==> y = x | y in A)

Singletons

We introduce an operation for forming singletons:

declare singleton: (S) [S] -> (Set S)

assert* singleton-def := [( singleton x = x ++ null)]

conclude singleton-characterization :=

(forall x y . x in singleton y <==> x = y)

pick-any x y

(! chain [(x in singleton y)

<==> (x in y ++ null) [singleton-def]

<==> (x = y | x in null) [in-def]

<==> (x = y | false) [NC]

<==> (x = y) [prop-taut ]])

Set operations

Next, we introduce three important binary set operations: union,

intersection, and difference:

declare union , intersection , diff:

(S) [(Set S) (Set S)] -> (Set S) [120 [alist- >set alist- >set]]

define [\/ /\ \] := [union intersection diff]

As we did with subset, we will define these operations constructively

and then deductively derive their classical nonconstructive

characterizations. We begin with union:

assert* union-def :=

[([] \/ s = s)

(h ++ t \/ s = h ++ (t \/ s))]

Set operations

We transform the output of eval so that sets are printed out in

square-bracket notation:

transform-output eval [set- >alist]

We use eval to test the definition:

> (eval [1 2 3] \/ [4 5 6 3])

List: [1 2 3 4 5 6]

The classical characterization of union is:

(forall A B x . x in A \/ B <==> x in A | x in B) (2)

Set operations

We derive (2) (with shortcut UC) using equivalence chaining:

conclude union-characterization :=

(forall A B x . x in A \/ B <==> x in A | x in B)

by-induction union-characterization {

null => pick-any B x

(! chain [(x in null \/ B)

<==> (x in B) [union-def]

<==> (false | x in B) [prop-taut]

<==> (x in null | x in B) [NC]])

| (A as (h insert t)) =>

let {IH := (forall B x . x in t \/ B <==> x in t | x in B)}

pick-any B x

(! chain [(x in A \/ B)

<==> (x in h ++ (t \/ B)) [union-def]

<==> (x = h | x in t \/ B) [in-def]

<==> (x = h | x in t | x in B) [IH]

<==> ((x = h | x in t) | x in B) [prop-taut]

<==> (x in A | x in B) [in-def ]])

Set operations

We continue with intersection:

assert* intersection-def :=

[(null /\ s = null)

(h ++ t /\ A = h ++ (t /\ A) <== h in A)

(h ++ t /\ A = t /\ A <== ~ h in A)]

> (eval [1 2 1] /\ [5 1 3])

List: [1]

The classical characterization of intersection follows:

(forall A B x . x in A /\ B <==> x in A & x in B) (3)

Set operations

We derive (3) in two steps by induction, with the basis case and

inductive case following:

conclude intersection-characterization-1 :=

(forall A B x . x in A /\ B ==> x in A & x in B)

by-induction intersection-characterization-1 {

null =>

pick-any B x

(! chain [(x in null /\ B)

==> (x in null) [intersection-def]

==> false [NC]

==> (x in null & x in B) [prop-taut ]])

| (A as (insert h t)) =>

...

}

Set operations

| (A as (insert h t)) =>

let {IH := (forall B x . x in t /\ B ==> x in t & x in B)}

pick-any B x

(! two-cases

assume (h in B)

(! chain

[(x in (h ++ t) /\ B)

==> (x in h ++ (t /\ B)) [intersection-def]

==> (x = h | x in t /\ B) [in-def]

==> (x = h | x in t & x in B) [IH]

==> ((x = h | x in t) & (x = h | x in B)) [prop-taut]

==> (x in A & (x in B | x in B)) [in-def (h in B)]

==> (x in A & x in B) [prop-taut ]])

assume (~ h in B)

(! chain [(x in A /\ B)

==> (x in t /\ B) [intersection-def]

==> (x in t & x in B) [IH]

==> (x in A & x in B) [in-def ]]))

Set operations

The two-cases analysis in the inductive case is dictated by the

corresponding case split in intersection-def.

The classical characterization of intersection, which we will abbreviate

as IC, is then readily derived from the two directions.

define intersection-characterization-2 :=

(forall A B x . x in A & x in B ==> x in A /\ B)

define intersection-characterization :=

(forall A B x . x in A /\ B <==> x in A & x in B)

define IC := intersection-characterization

Set operations

The following is an immediate corollary of IC:

conclude intersection-subset-theorem :=

(forall A B . A /\ B subset A)

pick-any A B

(! subset-intro

pick-any x

(! chain [(x in A /\ B)

==> (x in A) [IC]]))

Set operations

Finally, we go through the same process for the set-theoretic difference

operation:

assert* diff-def :=

[(null \ _ = null)

(h ++ t \ A = t \ A <== h in A)

(h ++ t \ A = h ++ (t \ A) <== ~ h in A)]

> (eval [1 2 3] \ [3 1])

List: [2]

CSCI.6964/4964 Software Verification—Fundamental Proof Methods in Computer Science (Arkoudas and Musser)—Sections 10.1-10.3.3 – p. 44/62



Set operations

We derive the following:

define diff-characterization-1 :=

(forall A B x . x in A \ B ==> x in A & ~ x in B)

define diff-characterization-2 :=

(forall A B x . x in A & ~ x in B ==> x in A \ B)

conclude diff-characterization :=

(forall A B x . x in A \ B <==> x in A & ~ x in B)

pick-any A B x

(! equiv

(! chain [(x in A \ B)

==> (x in A & ~ x in B) [diff-characterization-1 ]])

(! chain [(x in A & ~ x in B)

==> (x in A \ B) [diff-characterization-2 ]]))

define DC := diff-characterization

Set operations

We also introduce an operation that removes an element from a set,

using - as an alias for it:

declare remove: (S) [(Set S) S] -> (Set S) [- [alist- >set id]]

assert* remove-def :=

[(null - _ = null)

(h ++ t - x = t - x <== x = h)

(h ++ t - x = h ++ (t - x) <== x =/= h)]

(eval [1 2 3 2 5] - 2)

> List: [1 3 5]

Set operations

The corresponding characterization theorem for this operation follows:

define remove-characterization :=

(forall A x y . y in A - x <==> y in A & y =/= x)

define RC := remove-characterization

Some simple but useful corollaries of RC follow:

define remove-corollary := (forall A x . ~ x in A - x)

define remove-corollary-2 := (forall A x . ~ x in A ==> A - x = A)

define remove-corollary-3 :=

(forall A x y . x in A & y =/= x ==> x in A - y)

define remove-corollary-4 := (forall A x y . ~ x in A ==> ~ x in A - y)

define remove-corollary-5 :=

(forall A B x . A subset B & ~ x in A ==> A subset B - x)

Set operations

We have now derived characterization theorems for most of the

fundamental set-theoretic functions and relations: union, intersection,

set difference, the subset and identity relations, and removal.

We can proceed to prove some important results, starting with the

commutativity and associativity of union and intersection:

conclude intersection-commutes := (forall A B . A /\ B = B /\ A)

pick-any A B

(! set-identity-intro-direct

pick-any x

(! chain [(x in A /\ B) <==> (x in A & x in B) [IC]

<==> (x in B & x in A) [prop-taut]

<==> (x in B /\ A) [IC]]))

The proof of union commutativity is similar.

define union-commutes := (forall A B . A \/ B = B \/ A)

Set operations

We proceed with the associativity of intersection and union:

conclude intersection-associativity :=

(forall A B C . A /\ (B /\ C) = (A /\ B) /\ C)

pick-any A B C

(! set-identity-intro-direct

pick-any x

(! chain [(x in A /\ B /\ C)

<==> (x in A & x in B /\ C) [IC]

<==> (x in A & x in B & x in C) [IC]

<==> ((x in A & x in B) & x in C) [prop-taut]

<==> ((x in A /\ B) & x in C) [IC]

<==> (x in (A /\ B) /\ C) [IC]]))

define union-associativity := (forall A B C . A \/ B \/ C = (A \/ B) \/ C)

Set operations

Idempotence is also useful to derive for both operations:

conclude /\ -idempotence := (forall A . A /\ A = A)

pick-any A

(! set-identity-intro-direct

pick-any x

(! chain [(x in A /\ A)

<==> (x in A & x in A) [IC]

<==> (x in A) [prop-taut ]]))

Union idempotence is likewise established, and defined as

\/-idempotence.

Set operations

The following is a useful result specifying necessary and sufficient

conditions for the union of two sets to be empty:

conclude union-null-theorem :=

(forall A B . A \/ B = null <==> A = null & B = null)

pick-any A B

(! chain [(A \/ B = null)

<==> (forall x . x in A \/ B <==> x in null) [SIC]

<==> (forall x . x in A \/ B <==> false) [NC]

<==> (forall x . x in A | x in B <==> false) [UC]

<==> (forall x . ~ x in A & ~ x in B) [prop-taut]

<==> (( forall x . ~ x in A) & (forall x ~ x in B)) [quant-dist]

<==> (A = null & B = null) [NC-2 ]])

Set operations

We continue with two distributivity theorems. The first one shows that

union distributes over intersection:

∀ A,B,C . A ∪ (B ∩ C) = (A ∪B) ∩ (A ∪ C).

conclude distributivity-1 :=

(forall A B C . A \/ (B /\ C) = (A \/ B) /\ (A \/ C))

pick-any A B C

(! set-identity-intro-direct

pick-any x

(! chain [(x in A \/ (B /\ C))

<==> (x in A | x in B /\ C) [UC]

<==> (x in A | x in B & x in C) [IC]

<==> ((x in A | x in B) & (x in A | x in C)) [prop-taut]

<==> (x in A \/ B & x in A \/ C) [UC]

<==> (x in (A \/ B) /\ (A \/ C)) [IC]]))

Set operations

Our next distributivity result shows that intersection distributes over

union.

∀ A,B,C . A ∩ (B ∪ C) = (A ∩B) ∪ (A ∩ C)

define distributivity-2 :=

(forall A B C . A /\ (B \/ C) = (A /\ B) \/ (A /\ C))

Set operations

We continue with some theorems about set-theoretic difference:

conclude diff-theorem-1 := (forall A . A \ A = null)

pick-any A

(! set-identity-intro-direct

pick-any x

(! chain [(x in A \ A)

<==> (x in A & ~ x in A) [DC]

<==> false [prop-taut]

<==> (x in null) [NC]]))

Set operations

conclude diff-theorem-2 :=

(forall A B C . B subset C ==> A \ C subset A \ B)

pick-any A B C

assume (B subset C)

(! subset-intro

pick-any x

(! chain [(x in A \ C)

==> (x in A & ~ x in C) [DC]

==> (x in A & ~ x in B) [SC]

==> (x in A \ B) [DC]]))

Set operations

We go on to derive a number of additional results relating the three

fundamental set operations of union, intersection, and difference:

define diff-theorem-3 := (forall A B . A \ (A /\ B) = A \ B)

conclude diff-theorem-4 :=

(forall A B . A /\ (A \ B) = A \ B)

pick-any A B

(! set-identity-intro-direct

pick-any x

(! chain [(x in A /\ (A \ B))

<==> (x in A & x in A \ B) [IC]

<==> (x in A & x in A & ~ x in B) [DC]

<==> (x in A & ~ x in B) [prop-taut]

<==> (x in A \ B) [DC]]))

Set operations

define diff-theorem-5 := (forall A B . (A \ B) \/ B = A \/ B)

define diff-theorem-6 := (forall A B . (A \/ B) \ B = A \ B)

define diff-theorem-7 := (forall A B . (A /\ B) \ B = null)

conclude diff-theorem-8 :=

(forall A B . (A \ B) /\ B = null)

pick-any A B

(! set-identity-intro-direct

pick-any x

(! chain [(x in (A \ B) /\ B)

<==> (x in A \ B & x in B) [IC]

<==> ((x in A & ~ x in B) & x in B) [DC]

<==> false [prop-taut]

<==> (x in null) [NC]]))

define diff-theorem-9 :=

(forall A B C . A \ (B \/ C) = (A \ B) /\ (A \ C))

Set operations

conclude diff-theorem-10 :=

(forall A B C . A \ (B /\ C) = (A \ B) \/ (A \ C))

pick-any A B C

(! set-identity-intro-direct

pick-any x

(! chain [(x in A \ (B /\ C))

<==> (x in A & ~ x in B /\ C) [DC]

<==> (x in A & ~ (x in B & x in C)) [IC]

<==> (x in A & (~ x in B | ~ x in C)) [prop-taut]

<==> ((x in A & ~ x in B) | (x in A & ~ x in C)) [prop-taut]

<==> (x in A \ B | x in A \ C) [DC]

<==> (x in (A \ B) \/ (A \ C)) [UC]]))

define diff-theorem-11 := (forall A B . A \ (A \ B) = A /\ B)

define diff-theorem-12 := (forall A B . A subset B ==> A \/ (B \ A) = B)

Set operations

conclude subset-theorem-1 :=

(forall A B . A subset B ==> A \/ B = B)

pick-any A B

assume (A subset B)

(! set-identity-intro-direct

pick-any x

(! chain [(x in A \/ B)

<==> (x in A | x in B) [UC]

<==> (x in B | x in B) [prop-taut SC]

<==> (x in B) [prop-taut ]]))

define subset-theorem-2 := (forall A B . A subset B ==> A /\ B = A)

define intersection-subset-theorem' :=

(forall A B C . A subset B /\ C <==> A subset B & A subset C)

Set operations

conclude union-lemma-2 :=

(forall A B x . x ++ (A \/ B) = A \/ x ++ B)

pick-any A B x

(! chain [(x ++ (A \/ B))

= (x ++ (B \/ A)) [union-commutes]

= ((x ++ B) \/ A) [union-def]

= (A \/ (x ++ B)) [union-commutes ]])

Set operations

conclude absorption-1 :=

(forall x A . x in A <==> x ++ A = A)

pick-any x A

(! equiv

assume hyp := (x in A)

(! set-identity-intro-direct

pick-any y

(! chain [(y in x ++ A)

<==> (y = x | y in A) [in-def]

<==> (y in A | y in A) [hyp prop-taut]

<==> (y in A) [prop-taut ]]))

assume (x ++ A = A)

(!chain- > [true ==> (x in x ++ A) [in-lemma-1]

==> (x in A) [SIC ]]))

Note that on the second step of the equivalence chain, we specify both

hyp and prop-taut as justifiers. Neither would be sufficient by itself.

Set operations

The union analogue of intersection-subset-theorem’ does not

hold:

> (falsify (forall A B C . A subset B \/ C <==> A subset B | A subset C) 10)

List: ['success

|{?A:(Set Int) :=

insert 1

(insert 2

null:(Set Int )))

?B:(Set Int) := (insert 1

null:(Set Int))

?C:(Set Int) := (insert 2

null:(Set Int))

}|]

However, a weaker version does hold:

define union-subset-theorem :=

(forall A B C . A subset B | A subset C ==> A subset B \/ C)

